
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc99539250]Corrections required
0.1 [bookmark: _Toc99539251]Correction to f_NR_RRC_CipherActTime_GetCurrent ()
	Function name
	f_NR_RRC_CipherActTime_GetCurrent ()

	Reason for change
	It has been observed that some of the TCs are calling this function while cell is in switched off state and as per TS 36.523-3 section 7.4.5 cases where cells being switched off need not to be mapped to physical resources. Therefore, SS is not expected to fill the list if that cell is in switched off state which in turn leads to runtime error for v_SecurityActTimeList as it remains uninitialized for such cases.

	Summary of change
	Initialized variable v_SecurityActTimeList := {} so that behavior of such test cases can be handled here with empty list.


	TTCN module
	\Common\NR\NR_SecuritySteps.ttcn

	MCC160 Comment
	This will result in no bearers being ciphered when they should be.
Alternative solution implemented: ensured the cell power had been increased before activating security.  This affects 6.4.1.1, 6.4.2.1 and 6.4.2.2



Before Change:
	function f_NR_RRC_CipherActTime_GetCurrent(NR_CellId_Type p_NR_CellId,
                                             boolean p_IncrSRB1,
                                             boolean p_IncrSRB1_DL := false,
                                             template(omit) NR_PdcpCountInfoList_Type p_PdcpCountInfoList:= omit
                                             ) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type
  { //@sic R5s220123: added p_PdcpCountInfoList sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList;
    var NR_PdcpCountInfo_Type v_PdcpCountInfo;
    var NR_PDCP_ActTime_Type v_PDCP_ActTimeUL;
    var NR_PDCP_ActTime_Type v_PDCP_ActTimeDL;
    var NR_RadioBearerId_Type v_RadioBearerId;
    var NR_SecurityActTime_Type v_SecurityActTime;
    var integer I;
    var NR_SecurityActTimeList_Type v_SecurityActTimeList;
   
    if (isvalue(p_PdcpCountInfoList)) {
        v_PdcpCountInfoList := valueof(p_PdcpCountInfoList);
    } else {
       v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_NR_CellId);
    }
    
    for (I:=0; I < lengthof(v_PdcpCountInfoList); I:=I+1) {
      v_PdcpCountInfo := v_PdcpCountInfoList[I];
      if ((not ispresent(v_PdcpCountInfo.UL)) or (not ispresent(v_PdcpCountInfo.DL))) {
        FatalError(__FILE__, __LINE__, "missing PdcpCountInfo for UL or DL");
      }
      
      v_RadioBearerId := v_PdcpCountInfo.RadioBearerId;
      // Convert PdcpCount to integer value (UL an DL)
      v_PDCP_ActTimeUL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.UL);
      v_PDCP_ActTimeDL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.DL);
      
      // increment SQN for SRB1 (UL and DL)
      if (p_IncrSRB1 and (ischosen(v_RadioBearerId.Srb) and (v_RadioBearerId.Srb == tsc_SRB1))) { // @sic R5s190214 sic@
        v_PDCP_ActTimeUL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeUL.SQN);
        v_PDCP_ActTimeDL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeDL.SQN);
      }

      // When transmitting RRCReestablishment to UE on DL SRB1 ciphering is not applied
      // When receiving RRCReestablishemntComplete on UL SRB1 ciphering is applied by UE.
      // So the new ciphering keys needs to be applied at SN+1 in DL and at SN in UL => increment DL SN only
      // @sic R5s201221 sic@
      if ((p_IncrSRB1_DL) and (ischosen(v_RadioBearerId.Srb) and (v_RadioBearerId.Srb == tsc_SRB1))) {
        v_PDCP_ActTimeDL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeDL.SQN);
      }
      
      // set activation time
      v_SecurityActTime.RadioBearerId := v_RadioBearerId;
      v_SecurityActTime.UL := v_PDCP_ActTimeUL;
      v_SecurityActTime.DL := v_PDCP_ActTimeDL;
      
      v_SecurityActTimeList[I] := v_SecurityActTime;
    }
    return v_SecurityActTimeList;
  }







After Change:
	function f_NR_RRC_CipherActTime_GetCurrent(NR_CellId_Type p_NR_CellId,
                                             boolean p_IncrSRB1,
                                             boolean p_IncrSRB1_DL := false,
                                             template(omit) NR_PdcpCountInfoList_Type p_PdcpCountInfoList:= omit
                                             ) runs on NR_BASE_PTC return NR_SecurityActTimeList_Type
  { //@sic R5s220123: added p_PdcpCountInfoList sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList;
    var NR_PdcpCountInfo_Type v_PdcpCountInfo;
    var NR_PDCP_ActTime_Type v_PDCP_ActTimeUL;
    var NR_PDCP_ActTime_Type v_PDCP_ActTimeDL;
    var NR_RadioBearerId_Type v_RadioBearerId;
    var NR_SecurityActTime_Type v_SecurityActTime;
    var integer I;
    var NR_SecurityActTimeList_Type v_SecurityActTimeList := {}; // WA#WI=950222
   
    if (isvalue(p_PdcpCountInfoList)) {
        v_PdcpCountInfoList := valueof(p_PdcpCountInfoList);
    } else {
       v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_NR_CellId);
    }
    
    for (I:=0; I < lengthof(v_PdcpCountInfoList); I:=I+1) {
      v_PdcpCountInfo := v_PdcpCountInfoList[I];
      if ((not ispresent(v_PdcpCountInfo.UL)) or (not ispresent(v_PdcpCountInfo.DL))) {
        FatalError(__FILE__, __LINE__, "missing PdcpCountInfo for UL or DL");
      }
      
      v_RadioBearerId := v_PdcpCountInfo.RadioBearerId;
      // Convert PdcpCount to integer value (UL an DL)
      v_PDCP_ActTimeUL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.UL);
      v_PDCP_ActTimeDL.SQN := f_NR_SQN_ConvertFromPdcpCount(v_PdcpCountInfo.DL);
      
      // increment SQN for SRB1 (UL and DL)
      if (p_IncrSRB1 and (ischosen(v_RadioBearerId.Srb) and (v_RadioBearerId.Srb == tsc_SRB1))) { // @sic R5s190214 sic@
        v_PDCP_ActTimeUL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeUL.SQN);
        v_PDCP_ActTimeDL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeDL.SQN);
      }

      // When transmitting RRCReestablishment to UE on DL SRB1 ciphering is not applied
      // When receiving RRCReestablishemntComplete on UL SRB1 ciphering is applied by UE.
      // So the new ciphering keys needs to be applied at SN+1 in DL and at SN in UL => increment DL SN only
      // @sic R5s201221 sic@
      if ((p_IncrSRB1_DL) and (ischosen(v_RadioBearerId.Srb) and (v_RadioBearerId.Srb == tsc_SRB1))) {
        v_PDCP_ActTimeDL.SQN := f_NR_SQN_Increment(v_PDCP_ActTimeDL.SQN);
      }
      
      // set activation time
      v_SecurityActTime.RadioBearerId := v_RadioBearerId;
      v_SecurityActTime.UL := v_PDCP_ActTimeUL;
      v_SecurityActTime.DL := v_PDCP_ActTimeDL;
      
      v_SecurityActTimeList[I] := v_SecurityActTime;
    }
    return v_SecurityActTimeList;
  } 


MCC160 Implementation:
   function f_TC_6_4_1_1_NR5GC_TestBody() runs on NR5GC_PTC
   {// Power levels on FR2 are FFS
    // @sic R5s201047, R5-203382, R5s210094, R5-210062, R5s210454, R5-216995, R5-220543 sic@
    var template (value) NR_CellPowerList_Type v_CellPowerList_T0;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T2;
    var template (omit) NG_NAS_MSG_Indication_Type v_ReceivedMsgTemplate;
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var PLMN_Identity v_ASN_PLMN;
    var NAS_PlmnId v_PLMN;
    var NR_DRBInfoList_Type v_ExistingDRBs;
    var template(value) RadioBearerConfig v_RadioBearerConfig;
    var integer i := 0;
    var template(omit) DRB_ToAddModList v_DRB_ToAddModList := omit;
    var template (value) NR_RLC_BearerToAddModList_Type v_NR_RLC_BearerToAddModList;
    var template (value) CellGroupConfig v_SCGConfig;
    var template(value) SpCellConfig v_SpCellConfig;
    var NR_UplinkBWP_Type v_NR_UplinkBWP;
    timer t_Tmin;
    timer t_Tmax;

    v_CellPowerList_T0 := {
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2), // FR2 FFS
      cs_NR_CellPower (nr_Cell13, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };

    v_CellPowerList_T1 := {
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableCellSSS_EPRE_FR1, tsc_NR_NonSuitableCellSSS_EPRE_FR2), // FR2 FFS
      cs_NR_CellPower (nr_Cell13, -78, -78),
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };

    v_CellPowerList_T2 := {
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell13, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
      cs_NR_CellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };

    // @siclog "Step 0A" siclog@
    // SS adjusts cell levels according to row T0 of table 6.4.1.1.3.2-1/2
    f_NR_SetCellPowerList(v_CellPowerList_T0);

    // @siclog "Step 0B" siclog@
    // The UE is switched on
    f_UT_SwitchOnUE(UT);

    // @siclog "Step 0C1-0C19a1" siclog@
    // The steps 1-19a1 specified in generic procedure TS 38.508-1 Table 4.5.3.2-1 are performed
    v_ASN_PLMN := f_NR_CellInfo_GetPLMN (nr_Cell12);
    v_PLMN := f_NR_Asn2Nas_PlmnId(v_ASN_PLMN);
    v_ReceivedMsg := f_NR_RRC_ConnEst_DefWithNas (nr_Cell12, tsc_NR_RRC_TI_Def, ?, (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected));
    if (not f_Check_NG_RegistrationReqMsg (v_GMM_MobilityInfo, v_ReceivedMsg.Pdu, Initial_NoSecurity)) {
      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");
    }
       
    f_NR5GC_RRC_Idle_Steps5_13(v_GMM_MobilityInfo, nr_Cell12, v_ReceivedMsg, v_PLMN, TESTMode_OFF, Initial_NoSecurity);

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,
                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell12),
                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell12)),
                                                                                {bit2oct(f_NR_CellInfo_GetTAC(nr_Cell12))}),
                                                   cs_PLMN_List_1PLMN(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell13))));  // Table 6.4.1.1.3.3-3
    
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell12,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   cs_TimingInfo_Now,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
    f_NR5GC_RRC_Idle_Steps15_20(nr_Cell12, TESTMode_OFF, noRrcConnectionRelease);

    // @siclog "Step 0D" siclog@
    // SS starts timers of tmin = 2min and tmax = (6 minutes + 5min)
    t_Tmin.start(f_NR_SetTimerToleranceMin(nr_Cell12, nonProtocolTimer, 120.0));
    t_Tmax.start(f_NR_SetTimerToleranceMax(nr_Cell12, nonProtocolTimer, 660.0));

    // @siclog "Step 0Ea1-0F" siclog@
    // steps 19Aa1-20 specified in generic procedure TS 38.508-1 Table 4.5.3.2-1 are performed
    if (pc_noOf_PDUsNewConnection >0 ){
      f_NR_RRCRelease(nr_Cell12);
      f_NR5GC_RRC_Idle_Extension(nr_Cell12, TESTMode_OFF, noRrcConnectionRelease);
    }
    f_NR_RRC_InactiveDef(nr_Cell12, -, false);

    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell13 ,tsc_NR_38508_NextHopChainingCount); // @sic R5s200657 sic@

    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
    v_NR_RLC_BearerToAddModList[0] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_SRB(tsc_SRB2), omit, omit, cs_38508_LogicalChannelConfig_SRB(tsc_SRB2), true_);
    
    for (i:= 0; i < lengthof(v_ExistingDRBs); i := i+1) {
      v_DRB_ToAddModList[i] := cs_NR_DRB_ToAddMod(omit, v_ExistingDRBs[i].DRB_Id, true_, -, omit);
      v_NR_RLC_BearerToAddModList[i+1] := cs_NR_BearerConfigDef(f_NR_LogicalChannelId_DRB(v_ExistingDRBs[i].DRB_Id), omit, omit, omit, true_);
    }
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef( {cs_NR_SRB_ToAddMod_ReestablishPDCP(tsc_SRB2)}, v_DRB_ToAddModList);
    
    v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP (nr_Cell13, tsc_NR_BWP_Id);
    v_NR_UplinkBWP.Dedicated.R15.pusch_Config := omit;
    v_SpCellConfig := f_NR_GetSpCellConfigDef(nr_Cell13, cs_NR_ServingCellConfigDef(omit, 0, omit, cs_NR_UplinkConfigDef(v_NR_UplinkBWP.Dedicated.R15, omit)));
    v_SCGConfig := cs_38508_CellGroupConfig(f_NR_CellInfo_Get_CellGroupId(nr_Cell13), v_NR_RLC_BearerToAddModList, -, -, -, v_SpCellConfig);

    f_NR5GC_MobileInfo_SetSessionInfoList({}); // @sic R5s210407 sic@

    // @siclog "Step 1" siclog@
    // SS adjusts cell levels according to row T1 of table 6.4.1.1.3.2-1/2
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell13 ,tsc_NR_38508_NextHopChainingCount); // @sic R5s200657 sic@
    // @siclog "Step 2" siclog@
    // Check: Does the UE transmit an RRCResumeRequest message on NR Cell 13?
    f_NR_RRC_ResumeRequest_Def(nr_Cell13, -, mo_Signalling);
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");
...

  function f_TC_6_4_2_1_NR5GC_TestBody() runs on NR5GC_PTC
  { // @sic R5s200377 R5-204415 R5-205600 R5-212084 R5s210581 sic@
    //FR2 Power levels are FFS
    var integer v_T1_Cell1_RS_EPRE_FR1 := -115;
    var integer v_T2_Cell1_RS_EPRE_FR1 := tsc_NR_ServingCellSSS_EPRE_FR1;
    var integer v_T2_Cell2_RS_EPRE_FR1 := -78; // @sic R5-198879 sic@
    var integer v_T0_Cell1_RS_EPRE_FR1 := tsc_NR_ServingCellSSS_EPRE_FR1;
    var integer v_T0_Cell2_RS_EPRE_FR1 := tsc_NR_NonSuitableOffCellSSS_EPRE;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    
    var octetstring v_EncodedRrcSetup;
    var template(value) DL_CCCH_Message v_RrcSetup;
    
    var template(value) CellGroupConfig v_CellGroupConfig;
    var template(value) PDSCH_Config v_PDSCH_Config;
    var NR_DownlinkBWP_Type v_NR_DownlinkBWP_Cell2;
    var template(omit) NR_BWP_DownlinkDedicated_PDSCH_Config_Type v_BWP_DownlinkDedicated_PDSCH_Config;
    var integer v_SSB_IndexCell2;
    var template (value) PUSCH_Config v_PUSCH_Config;

    v_RrcSetup := f_NR_38508_RRCSetup(nr_Cell1);
    v_EncodedRrcSetup := f_NR_RRC_DL_CCCH_Encvalue(v_RrcSetup);
    

    v_CellPowerList_AtT0:={
      cs_NR_CellPower(nr_Cell1, v_T0_Cell1_RS_EPRE_FR1, v_T0_Cell1_RS_EPRE_FR1),
      cs_NR_CellPower(nr_Cell2, v_T0_Cell2_RS_EPRE_FR1, v_T0_Cell2_RS_EPRE_FR1)
    };
    v_CellPowerList_AtT1:={
      cs_NR_CellPower(nr_Cell1, v_T1_Cell1_RS_EPRE_FR1, v_T1_Cell1_RS_EPRE_FR1)
    };
    v_CellPowerList_AtT2:={
      cs_NR_CellPower(nr_Cell1, v_T2_Cell1_RS_EPRE_FR1, v_T2_Cell1_RS_EPRE_FR1),
      cs_NR_CellPower(nr_Cell2, v_T2_Cell2_RS_EPRE_FR1, v_T2_Cell2_RS_EPRE_FR1)
    };

    // @siclog "Step 1" siclog@
    // The SS changes NR Cell 1 SSS level according to the row "T1" in table 6.4.2.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    // @siclog "Step 2 - Void" siclog@
    // @siclog "Step 2A" siclog@
    // Wait 15s (Note 1)
    f_Delay(15.0);
    
    // @siclog "Step 3" siclog@
    // The SS transmits a Paging message.
    f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

    // @siclog "Step 4" siclog@
    // Check: Does the UE attempt to transmit an uplink message within the next 10s?
    f_NR_Check_NoUplinkMessage(10.0);

    // @siclog "Step 5" siclog@
    // The UE is switched off.
    f_UT_SwitchOffUE(UT, false);

    // @siclog "Steps 6/6A - Void" siclog@
    // @siclog "Step 6B" siclog@
    // The SS changes NR Cell 1 SSS levels according to the row "T0" in table 6.4.2.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT0);
    
    f_NR_SS_RachProcedureConfig(nr_Cell1, cs_NR_RachProcedureConfig_CBRA_Msg4Based( f_NR_CellInfo_GetPhysicalParameters(nr_Cell1), cs_NR_RachProcedureMsg4WithCcchMsg(v_EncodedRrcSetup)));

    // @siclog "Step 6C" siclog@
    // The UE is switched on.
    f_UT_SwitchOnUE(UT, false);
    
    f_NR5GC_MobileInfo_Init();// @sic R5s210092 sic@
    
    // @siclog "Step 7" siclog@
    // Generic procedure specified in TS 38.508-1 Table 4.5.3.2-1 are performed, in which the RRCRelease message includes the suspendConfig with NR_RRC_INACTIVE condition thereby moving the UE to RRC_INACTIVE.
    f_NR5GC_RRC_IdleState1N_Def(nr_Cell1, -, noRrcConnectionRelease, Initial_NoSecurity); // @sic R5s210092 sic@
    f_NR_RRC_InactiveDef(nr_Cell1, -, false);
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell2,tsc_NR_38508_NextHopChainingCount);
    f_NR_SS_RachProcedureConfig(nr_Cell2, cs_NR_RachProcedureConfig_CBRA_Msg4Based(f_NR_CellInfo_GetPhysicalParameters(nr_Cell2), cs_NR_RachProcedureMsg4WithNoRrcMsg));
    
    // @siclog "Step 8-15 Void" siclog@
    // @siclog "Step 16" siclog@
    // The SS changes NR Cell 2 SSS level according to the row "T2" in table 6.4.2.1.3.2-1/2.
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell2,tsc_NR_38508_NextHopChainingCount);
    // @siclog "Step 17 - Void" siclog@
    // @siclog "Step 18" siclog@
    // Check: Does the UE transmit an RRCResumeRequest message on the NR Cell 2?
    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, rna_Update)));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 18");

    // @siclog "Step 18A" siclog@
    // The SS transmits an RRCResume message
    v_SSB_IndexCell2 := f_NR_CellInfo_GetSSB_Index(nr_Cell2);

    v_NR_DownlinkBWP_Cell2 := f_NR_CellInfo_GetDownlinkBWP(nr_Cell2, tsc_NR_BWP_Id);
    v_PDSCH_Config := v_NR_DownlinkBWP_Cell2.Pdsch.ConfigDedicated.R15; // Table 6.4.2.1.3.3-6
    v_PDSCH_Config.dmrs_DownlinkForPDSCH_MappingTypeA := omit;
    v_PDSCH_Config.tci_StatesToAddModList[0] := cs_38508_TCI_State(v_SSB_IndexCell2);
    v_PDSCH_Config.prb_BundlingType := cs_NR_Prb_OmitBundleSize;
    v_BWP_DownlinkDedicated_PDSCH_Config := {setup := v_PDSCH_Config};
    v_CellGroupConfig := f_NR_GetSCGConfigResume(nr_Cell2);
    v_CellGroupConfig.spCellConfig.spCellConfigDedicated.initialDownlinkBWP := cs_BWP_DownlinkDedicated_Def(-, v_BWP_DownlinkDedicated_PDSCH_Config);
    
    v_PUSCH_Config := cs_38508_PUSCH_Config_FR1(cs_38508_DMRS_UplinkConfig(pos1), v_SSB_IndexCell2);
    v_CellGroupConfig.spCellConfig.spCellConfigDedicated.uplinkConfig.initialUplinkBWP.pusch_Config := {setup := v_PUSCH_Config}; // @sic R5s210092 sic@
    

    f_NR_SendRRCResume_Def(nr_Cell2, -, -, v_CellGroupConfig);
    
    // @siclog "Step 18B" siclog@
    // The UE transmit an RRCResumeComplete message
    f_NR_RRCResumeComplete_Def(nr_Cell2);

    // @siclog "Step 19" siclog@
    // The SS transmits an RRCRelease message including suspendConfig with NR_RRC_INACTIVE condition.
    f_NR_RRC_InactiveDef(nr_Cell2);

  }//end of f_TC_6_4_2_1_NR5GC_TestBody
  
  function f_TC_6_4_2_2_NR5GC_TestBody() runs on NR_BASE_PTC
  { // @sic R5s200403 R5-203502 sic@
    //FR2 Power levels are FFS
    var integer v_T1_Cell3_RS_EPRE_FR1 := -78; // @sic R5-198879 sic@
    var integer v_T2_Cell1_RS_EPRE_FR1 := tsc_NR_NonSuitableOffCellSSS_EPRE;
    var integer v_T2_Cell6_RS_EPRE_FR1 := -78;
    var integer v_T3_Cell6_RS_EPRE_FR1 := -98;

    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT3;

    v_CellPowerList_AtT1:={
      cs_NR_CellPower(nr_Cell3, v_T1_Cell3_RS_EPRE_FR1, v_T1_Cell3_RS_EPRE_FR1)
    };
    v_CellPowerList_AtT2:={
      cs_NR_CellPower(nr_Cell1, v_T2_Cell1_RS_EPRE_FR1, v_T2_Cell1_RS_EPRE_FR1),
      cs_NR_CellPower(nr_Cell6, v_T2_Cell6_RS_EPRE_FR1, v_T2_Cell6_RS_EPRE_FR1)
    };
    v_CellPowerList_AtT3:={
      cs_NR_CellPower(nr_Cell6, v_T3_Cell6_RS_EPRE_FR1, v_T3_Cell6_RS_EPRE_FR1)
    };

    // @siclog "Step 1" siclog@
    // The SS changes NR Cell 3 SSS level according to the row "T1" in table 6.4.2.2.3.2-1/2.
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell3);
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell3);
    // @siclog "Step 2 - Void" siclog@

    // @siclog "Step 3" siclog@
    // Check: Does the UE transmit an RRCResumeRequest message on the cell 3?
    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell3, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, rna_Update)));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    // @siclog "Step 4" siclog@
    // The SS transmits an RRCRelease message including suspendConfig with NR_RRC_INACTIVE condition.
    f_NR_RRC_InactiveDef(nr_Cell3, -, false);

    // @siclog "Step 5" siclog@
    // The SS changes NR Cell 1 and NR Cell 6 SSS levels according to the row "T2" in table 6.4.2.2.3.2-1/2.
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell6);
    f_NR_SetCellPowerList(v_CellPowerList_AtT2);
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell6);
    // @siclog "Step 6 - Void" siclog@

    // @siclog "Step 7" siclog@
    // Check: Does the UE transmit an RRCResumeRequest message on the cell 6?
    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell6, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, rna_Update)));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // @siclog "Step 8" siclog@
    // The SS transmits an RRCRelease message including suspendConfig with NR_RRC_INACTIVE condition.
    f_NR_RRC_InactiveDef(nr_Cell6,-,false);

    // @siclog "Step 9" siclog@
    // The SS changes NR Cell 6 SSS level according to the row "T3" in table 6.4.2.2.3.2-1/2.
    f_NR_SS_AS_ActivateSecurity_Current(nr_Cell3);
    f_NR_SetCellPowerList(v_CellPowerList_AtT3);

    // @siclog "Step 10 - Void" siclog@

    // @siclog "Step 11" siclog@
    // Check: Does the UE transmit an RRCResumeRequest message on the cell 3?
    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell3, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, rna_Update)));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    // @siclog "Step 12" siclog@
    // The SS transmits an RRCRelease message including suspendConfig with NR_RRC_INACTIVE condition.
    f_NR_RRC_InactiveDef(nr_Cell3);

  }//end of f_TC_6_4_2_2_NR5GC_TestBody
  
